
======================================================================

The following is another MicroSoft BASIC program by Dennis Brothers.  It
demonstrates the Macintosh's sound generator.

1000 CLS: PRINT
1010 PRINT "Macintosh Four-Tone Synthesizer Music Demonstration"
1020 PRINT "            Theme from the New World"
1030 PRINT "           Version 1.00 - 7 May, 1984 -"
1031 PRINT "         Written for A+ Magazine (Issue ?)"
1040 PRINT "      Dennis F. Brothers - Compuserve 70065,172 -"
1041 PRINT "               MCI Mail DBROTHERS"
1050 PRINT
1060 PRINT "              Copyright (c) 1984 -"
1061 PRINT "         Brothers Associates, Wayland MA"
1070 PRINT "Permission is hereby granted for personal, non-"
1071 PRINT "commercial reproduction and use of this program,"
1072 PRINT "provided that this notice is included in any copy."
1100 PRINT
1110 PRINT "Initializing - Please wait....."
1120 CLEAR ,20000
1130 DEFINT A-Z  '  Default to integers for speed and space
1140 DIM PARAMLIST(49\2, 1)
1141 REM  '  Standard ROM Write parameter lists
1150 DIM SYNTHREC(2, 1)  ' Mode word and pointer to sound record
1160 DIM SOUNDREC(24,1)
1161 REM   Duration wrd, 4 long-wrd triplets (tone, phase, wave)
1170 DIM DURLIST(99), TONELIST(1,3,99)
1171 REM  '  lists of durations and tones to play
1180 GOSUB 8000
1181 REM   '  Go initialize the machine-language subroutines
1190 REM
2000 REM  Build the tone frequency table
2010 REM
2020 PRINT: PRINT "    Building tone tables....."
2030 DIM NOTEFREQ#(87)  '  Simulate a piano keyboard (88 keys)
2040 HALFSTEP#=2#^(1#/12#)  '  ratio of musical half-step
2050 NOTEFREQ#(0)=440#/16#

2052 REM '  4th A below middle C (A above middle C = 440 Hz)
2060 FOR I=1 TO 11
2070 NOTEFREQ#(I)=NOTEFREQ#(I-1)*HALFSTEP#
2071 REM  '  fill in base octave
2080 NEXT I
2090 FOR I=12 TO 87
2100 NOTEFREQ#(I)=NOTEFREQ#(I-12)*2#  '  fill in rest of array
2110 NEXT I
2120 REM
2130 REM  Set up table of synthesizer step rates corresponding
2131 REM  to frequencies
2140 DIM NOTERATE(1,87)
2150 STEPSPERHZ#=(256#*(704#/15667200#))
2160 FOR I=0 TO 87
2170 RATE#=NOTEFREQ#(I)*STEPSPERHZ#*65536#
2180 LPOKE!=FNCODE!(LPOKE):
     CALL LPOKE!(VARPTR(NOTERATE(0,I)),RATE#)
2190 NEXT I
2200 REM
3000 REM  Build various waveform tables
3010 REM
3020 PRINT: PRINT "    Building waveform tables....."
3030 REM
3040 REM  Set up a square-wave waveform table
3050 DIM WAVESQUARE(127)
3060 FOR I=0 TO 127
3070 POKE VARPTR(WAVESQUARE(0))+I,255
3080 NEXT I
3090 FOR I=128 TO 255
3100 POKE VARPTR(WAVESQUARE(0))+I,0
3110 NEXT I
3120 REM
3130 REM  Set up a triangle-wave waveform table
3140 REM
3150 DIM WAVETRIANG(127)
3160 FOR I=0 TO 63
3170 POKE VARPTR(WAVETRIANG(0))+I,128+(I*2)
3180 NEXT I
3190 FOR I=64 TO 191
3200 POKE VARPTR(WAVETRIANG(0))+I,255-((I-64)*2)
3210 NEXT I
3220 FOR I=192 TO 255
3230 POKE VARPTR(WAVETRIANG(0))+I,(I-192)*2
3240 NEXT I
3250 REM
3260 REM  Set up a sine-wave waveform table
3270 REM
3280 DIM WAVESINE(127)
3290 PI#=3.14159265#
3300 FOR I=0 TO 255
3310 AMPLITUDE#=SIN((I/256#)*2#*PI#)
3320 POKE VARPTR(WAVESINE(0))+I,FIX(127.5#+AMPLITUDE#*127#)
3330 NEXT I



3340 REM
3350 REM  Set up a fundamental plus third-harmonic waveform
3351 REM  table
3360 REM
3370 DIM WAVETHIRD(127)
3380 FOR I=0 TO 255
3390 AMPLITUDE#=(SIN((I/256#)*2#*PI#)+SIN(((I*3)/256#)*2#*PI#))
3391 AMPLITUDE#=AMPLITUDE#/2#
3400 POKE VARPTR(WAVETHIRD(0))+I,FIX(127.5#+AMPLITUDE#*127#)
3410 NEXT I

