
======================================================================

7180 REM
8000 REM  Initialize the machine language subroutines
8010 REM
8020 REM HEX Intel format data for machine-language subroutines
8030 DATA "1C000000000C002000300044005400624E560000206E00083010206E000C3080CA"
8040 DATA "1C001C004E5E4E754E560000206E000A30AE00084E5E4E754E560000206E00088E"
8050 DATA "1C0038002010206E000C20804E5E4E754E560000206E000C20AE00084E5E4E7550"
8060 DATA "1C0054004E560000206E0008A0034E5E4E754E560000206E0008A4034E5E4E75F4"
8070 DATA "000000000000"
8080 REM
8090 DIM CODEARRAY(200)
8100 DEF FNCODE!(OFFSET)=VARPTR(CODEARRAY(0))+OFFSET
8110 RESTORE 8030
8111 REM  '  Restore to machine-language HEX data statements
8120 MLLINE=8030  '  First DATA statement line number
8130 READ ML$  '  Read next line of HEX data
8140 MLCHK=0  '  Initialize checksum
8150 FOR I=1 TO LEN(ML$)-1 STEP 2
8151 REM  '  Scan by bytes (pairs of hex digits)
8160 MLCHK=(MLCHK+VAL("&H"+MID$(ML$,I,2))) MOD &H100
8161 REM'  Compute checksum
8170 NEXT I
8180 IF MLCHK<>0 THEN CLS:
     PRINT CHR$(7);"Error in ML line";MLLINE: STOP
8190 MLL=VAL("&H"+MID$(ML$,3,2)+MID$(ML$,1,2))
8191 REM   '  Get byte count of ML string
8200 MLS=VAL("&H"+MID$(ML$,7,2)+MID$(ML$,5,2))
8201 REM   '  Get start addr of ML string
8210 IF MLL=0 THEN GOTO 8280  '  Zero byte count ends ML data
8220 FOR I=0 TO MLL-1
8230 POKE FNCODE!(MLS+I),VAL("&H"+MID$(ML$,9+I*2,2))
8231 REM  '  Put ML data in array
8240 NEXT I
8250 MLLINE=MLLINE+10
8251 REM   '  Keep track of DATA line number, in case of error
8260 GOTO 8130

8270 REM  Set up CALL address pointers - offsets are at
8271 REM  beginning of array
8280 WPEEK=CODEARRAY(0): WPEEK!=0  '  Two-byte peek
8290 WPOKE=CODEARRAY(1): WPOKE!=0  '  Two-byte poke
8300 LPEEK=CODEARRAY(2): LPEEK!=0  '  Four-byte peek
8310 LPOKE=CODEARRAY(3): LPOKE!=0  '  Four-byte poke
8320 IOWRITE=CODEARRAY(4): IOWRITE!=0
8321 REM   '  Synchronous Write ROM call
8330 IOWRITEASYNC=CODEARRAY(5): IOWRITEASYNC!=0
8331 REM  '  Asynchronous Write ROM call
8340 RETURN

